

# NAG Library Routine Document

## F08KDF (DGESDD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08KDF (DGESDD) computes the singular value decomposition (SVD) of a real  $m$  by  $n$  matrix  $A$ , optionally computing the left and/or right singular vectors, by using a divide-and-conquer method.

### 2 Specification

```
SUBROUTINE F08KDF (JOBZ, M, N, A, LDA, S, U, LDU, VT, LDVT, WORK, LWORK,      &
                  IWORK, INFO)
INTEGER          M, N, LDA, LDU, LDVT, LWORK, IWORK(8*min(M,N)), INFO
REAL (KIND=nag_wp) A(LDA,*), S(min(M,N)), U(LDU,*), VT(LDVT,*),      &
                  WORK(max(1,LWORK))
CHARACTER(1)     JOBZ
```

The routine may be called by its LAPACK name *dgesdd*.

### 3 Description

The SVD is written as

$$A = U\Sigma V^T,$$

where  $\Sigma$  is an  $m$  by  $n$  matrix which is zero except for its  $\min(m,n)$  diagonal elements,  $U$  is an  $m$  by  $m$  orthogonal matrix, and  $V$  is an  $n$  by  $n$  orthogonal matrix. The diagonal elements of  $\Sigma$  are the singular values of  $A$ ; they are real and non-negative, and are returned in descending order. The first  $\min(m,n)$  columns of  $U$  and  $V$  are the left and right singular vectors of  $A$ .

Note that the routine returns  $V^T$ , not  $V$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Arguments

- |  |   |
|--|---|
| 1: JOBZ – CHARACTER(1)   | <i>Input</i>  |
| <i>On entry:</i> specifies options for computing all or part of the matrix $U$ . |   |
| JOBZ = 'A'   | All $m$ columns of $U$ and all $n$ rows of $V^T$ are returned in the arrays U and VT.                             |
| JOBZ = 'S'   | The first $\min(m,n)$ columns of $U$ and the first $\min(m,n)$ rows of $V^T$ are returned in the arrays U and VT. |

**JOBZ = 'O'**

If  $M \geq N$ , the first  $n$  columns of  $U$  are overwritten on the array A and all rows of  $V^T$  are returned in the array VT. Otherwise, all columns of  $U$  are returned in the array U and the first  $m$  rows of  $V^T$  are overwritten in the array VT.

**JOBZ = 'N'**

No columns of  $U$  or rows of  $V^T$  are computed.

*Constraint:*  $\text{JOBZ} = \text{'A'}$ ,  $\text{'S'}$ ,  $\text{'O'}$  or  $\text{'N'}$ .

2:  $M$  – INTEGER *Input*

*On entry:*  $m$ , the number of rows of the matrix  $A$ .

*Constraint:*  $M \geq 0$ .

3:  $N$  – INTEGER *Input*

*On entry:*  $n$ , the number of columns of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

4:  $A(\text{LDA}, *)$  – REAL (KIND=nag\_wp) array *Input/Output*

**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .

*On entry:* the  $m$  by  $n$  matrix  $A$ .

*On exit:* if  $\text{JOBZ} = \text{'O'}$ , A is overwritten with the first  $n$  columns of  $U$  (the left singular vectors, stored column-wise) if  $M \geq N$ ; A is overwritten with the first  $m$  rows of  $V^T$  (the right singular vectors, stored row-wise) otherwise.

If  $\text{JOBZ} \neq \text{'O'}$ , the contents of A are destroyed.

5:  $\text{LDA}$  – INTEGER *Input*

*On entry:* the first dimension of the array A as declared in the (sub)program from which F08KDF (DGESDD) is called.

*Constraint:*  $\text{LDA} \geq \max(1, M)$ .

6:  $S(\min(M, N))$  – REAL (KIND=nag\_wp) array *Output*

*On exit:* the singular values of  $A$ , sorted so that  $S(i) \geq S(i + 1)$ .

7:  $U(\text{LDU}, *)$  – REAL (KIND=nag\_wp) array *Output*

**Note:** the second dimension of the array U must be at least  $\max(1, M)$  if  $\text{JOBZ} = \text{'A'}$  or  $\text{JOBZ} = \text{'O'}$  and  $M < N$ ,  $\max(1, \min(M, N))$  if  $\text{JOBZ} = \text{'S'}$ , and at least 1 otherwise.

*On exit:*

If  $\text{JOBZ} = \text{'A'}$  or  $\text{JOBZ} = \text{'O'}$  and  $M < N$ , U contains the  $m$  by  $m$  orthogonal matrix  $U$ .

If  $\text{JOBZ} = \text{'S'}$ , U contains the first  $\min(m, n)$  columns of  $U$  (the left singular vectors, stored column-wise).

If  $\text{JOBZ} = \text{'O'}$  and  $M \geq N$ , or  $\text{JOBZ} = \text{'N'}$ , U is not referenced.

8:  $\text{LDU}$  – INTEGER *Input*

*On entry:* the first dimension of the array U as declared in the (sub)program from which F08KDF (DGESDD) is called.

*Constraints:*

if  $\text{JOBZ} = \text{'S'}$  or  $\text{'A'}$  or  $\text{JOBZ} = \text{'O'}$  and  $M < N$ ,  $\text{LDU} \geq \max(1, M)$ ;  
otherwise  $\text{LDU} \geq 1$ .

9:      $\text{VT}(\text{LDVT}, *)$  – REAL (KIND=nag\_wp) array *Output*

**Note:** the second dimension of the array VT must be at least  $\max(1, N)$  if  $\text{JOBZ} = \text{'A'}$  or  $\text{'S'}$  or  $\text{JOBZ} = \text{'O'}$  and  $M \geq N$ , and at least 1 otherwise.

*On exit:* if  $\text{JOBZ} = \text{'A'}$  or  $\text{JOBZ} = \text{'O'}$  and  $M \geq N$ , VT contains the  $n$  by  $n$  orthogonal matrix  $V^T$ .

If  $\text{JOBZ} = \text{'S'}$ , VT contains the first  $\min(m, n)$  rows of  $V^T$  (the right singular vectors, stored row-wise).

If  $\text{JOBZ} = \text{'O'}$  and  $M < N$ , or  $\text{JOBZ} = \text{'N'}$ , VT is not referenced.

10:     $\text{LDVT}$  – INTEGER *Input*

*On entry:* the first dimension of the array VT as declared in the (sub)program from which F08KDF (DGESDD) is called.

*Constraints:*

if  $\text{JOBZ} = \text{'A'}$  or  $\text{JOBZ} = \text{'O'}$  and  $M \geq N$ ,  $\text{LDVT} \geq \max(1, N)$ ;  
 if  $\text{JOBZ} = \text{'S'}$ ,  $\text{LDVT} \geq \max(1, \min(M, N))$ ;  
 otherwise  $\text{LDVT} \geq 1$ .

11:     $\text{WORK}(\max(1, \text{LWORK}))$  – REAL (KIND=nag\_wp) array *Workspace*

12:     $\text{LWORK}$  – INTEGER *Input*

*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08KDF (DGESDD) is called.

If  $\text{LWORK} = -1$ , a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

*Suggested value:* for optimal performance, LWORK should generally be larger. Consider increasing LWORK by at least  $nb \times \min(M, N)$ , where nb is the optimal **block size**.

*Constraints:*

if  $\text{JOBZ} = \text{'N'}$ ,  $\text{LWORK} \geq 3 \times \min(M, N) + \max(1, \max(M, N), 7 \times \min(M, N))$ ;  
 if  $\text{JOBZ} = \text{'O'}$ ,  $\text{LWORK} \geq 3 \times \min(M, N) + \max(1, \max(M, N), 5 \times \min(M, N) \times \min(M, N) + 4 \times \min(M, N))$ ;  
 if  $\text{JOBZ} = \text{'S'}$  or  $\text{'A'}$ ,  $\text{LWORK} \geq 3 \times \min(M, N) + \max(1, \max(M, N), 4 \times \min(M, N) \times \min(M, N) + 4 \times \min(M, N))$ ;  
 otherwise  $\text{LWORK} \geq 1$ .

13:     $\text{IWORK}(8 \times \min(M, N))$  – INTEGER array *Workspace*

14:     $\text{INFO}$  – INTEGER *Output*

*On exit:*  $\text{INFO} = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

$\text{INFO} < 0$

If  $\text{INFO} = -i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\text{INFO} > 0$

F08KDF (DGESDD) did not converge, the updating process failed.

## 7 Accuracy

The computed singular value decomposition is nearly the exact singular value decomposition for a nearby matrix ( $A + E$ ), where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and  $\epsilon$  is the ***machine precision***. In addition, the computed singular vectors are nearly orthogonal to working precision. See Section 4.9 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F08KDF (DGESDD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08KDF (DGESDD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately proportional to  $mn^2$  when  $m > n$  and  $m^2n$  otherwise.

The singular values are returned in descending order.

The complex analogue of this routine is F08KRF (ZGESDD).

## 10 Example

This example finds the singular values and left and right singular vectors of the 4 by 6 matrix

$$A = \begin{pmatrix} 2.27 & 0.28 & -0.48 & 1.07 & -2.35 & 0.62 \\ -1.54 & -1.67 & -3.09 & 1.22 & 2.93 & -7.39 \\ 1.15 & 0.94 & 0.99 & 0.79 & -1.45 & 1.03 \\ -1.94 & -0.78 & -0.21 & 0.63 & 2.30 & -2.57 \end{pmatrix},$$

together with approximate error bounds for the computed singular values and vectors.

The example program for F08KBF (DGESVD) illustrates finding a singular value decomposition for the case  $m \geq n$ .

### 10.1 Program Text

```
Program f08kdfe

!     F08KDF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
Use nag_library, Only: ddisna, dgesdd, nag_wp, x02ajf, x04caf
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Integer, Parameter :: nb = 64, nin = 5, nout = 6
!     .. Local Scalars ..
Real (Kind=nag_wp) :: eps, serrbd
Integer :: i, ifail, info, lda, ldu, lwork, m, n
!     .. Local Arrays ..
```

```

Real (Kind=nag_wp), Allocatable :: a(:,:,), rcondu(:, ), rcondv(:, ), s(:, ), &
                                  u(:,:, ), uerrbd(:, ), verrbd(:, ), &
                                  work(:)
Real (Kind=nag_wp) :: dummy(1,1)
Integer, Allocatable :: iwork(:)
! .. Intrinsic Procedures ..
Intrinsic :: max, min, nint
! .. Executable Statements ..
Write (nout,*) 'F08KDF Example Program Results'
Write (nout,*)
! Skip heading in data file
Read (nin,*) 
Read (nin,*) m, n
lda = m
ldu = m
Allocate (a(lda,n),rcondu(m),rcondv(m),s(m),u(ldu,m),uerrbd(m), &
          verrbd(m),iwork(8*min(m,n)))
! Use routine workspace query to get optimal workspace.
lwork = -1
! The NAG name equivalent of dgesdd is f08kdf
Call dgesdd('Overwrite A by transpose(V)',m,n,a,lda,s,u,ldu,dummy,1, &
            dummy,lwork,iwork,info)
! Make sure that there is enough workspace for block size nb.
lwork = max((5*m+9)*m+n+nb*(m+n),nint(dummy(1,1)))
Allocate (work(lwork))
! Read the m by n matrix A from data file
Read (nin,*)(a(i,1:n),i=1,m)
! Compute the singular values and left and right singular vectors
! of A (A = U*S*(V**T), m.le.n)
! The NAG name equivalent of dgesdd is f08kdf
Call dgesdd('Overwrite A by transpose(V)',m,n,a,lda,s,u,ldu,dummy,1,work, &
            lwork,iwork,info)

If (info==0) Then
    ! Print solution
    Write (nout,*) 'Singular values'
    Write (nout,99999) s(1:m)
    Flush (nout)

    ! Normalize so that u(1,j)>=0
    Do i = 1, m
        If (u(1,i)<0.0_nag_wp) Then
            u(1:m,i) = -u(1:m,i)
            a(i,1:n) = -a(i,1:n)
        End If
    End Do
    ! ifail: behaviour on error exit
    !         =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
    ifail = 0
    Call x04caf('General',' ',m,m,u,ldu,'Left singular vectors',ifail)

    Write (nout,*) 
    Flush (nout)

    Call x04caf('General',' ',m,n,a,lda,'Right singular vectors by row '// &
                 '(first m rows of V**T)',ifail)

    ! Get the machine precision, EPS and compute the approximate
    ! error bound for the computed singular values. Note that for
    ! the 2-norm, S(1) = norm(A)
    eps = x02ajf()
    serrbd = eps*s(1)

```

```

!
!      Call DDISNA (F08FLF) to estimate reciprocal condition
!      numbers for the singular vectors

      Call ddisna('Left',m,n,s,rcondu,info)
      Call ddisna('Right',m,n,s,rcondv,info)

!
!      Compute the error estimates for the singular vectors

      Do i = 1, m
         uerrbd(i) = serrbd/rcondu(i)
         verrbd(i) = serrbd/rcondv(i)
      End Do

!
!      Print the approximate error bounds for the singular values
!      and vectors

      Write (nout,*)
      Write (nout,*) 'Error estimate for the singular values'
      Write (nout,99998) serrbd
      Write (nout,*)
      Write (nout,*) 'Error estimates for the left singular vectors'
      Write (nout,99998) uerrbd(1:m)
      Write (nout,*)
      Write (nout,*) 'Error estimates for the right singular vectors'
      Write (nout,99998) verrbd(1:m)
      Else
         Write (nout,99997) 'Failure in DGESDD. INFO =', info
      End If

99999 Format (3X,(8F8.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4)
End Program f08kdfe

```

## 10.2 Program Data

F08KDF Example Program Data

```

4       6                               :Values of M and N

2.27   0.28  -0.48   1.07  -2.35   0.62
-1.54  -1.67  -3.09   1.22   2.93  -7.39
 1.15   0.94   0.99   0.79  -1.45   1.03
-1.94  -0.78  -0.21   0.63   2.30  -2.57  :End of matrix A

```

## 10.3 Program Results

F08KDF Example Program Results

```

Singular values
 9.9966  3.6831  1.3569  0.5000
Left singular vectors
      1       2       3       4
1  0.1921  0.8030  0.0041  0.5642
2  -0.8794  0.3926 -0.0752 -0.2587
3   0.2140  0.2980  0.7827 -0.5027
4  -0.3795 -0.3351  0.6178  0.6017

Right singular vectors by row (first m rows of V**T)
      1       2       3       4       5       6
1   0.2774  0.2020  0.2918 -0.0938 -0.4213  0.7816
2   0.6003  0.0301 -0.3348  0.3699 -0.5266 -0.3353
3  -0.1277  0.2805  0.6453  0.6781  0.0413 -0.1645
4  -0.1323 -0.7034 -0.1906  0.5399  0.0575  0.3957

Error estimate for the singular values
 1.1E-15

```

```
Error estimates for the left singular vectors
 1.8E-16    4.8E-16    1.3E-15    1.3E-15

Error estimates for the right singular vectors
 1.8E-16    4.8E-16    1.3E-15    2.2E-15
```

---